17

Inventory, Production, and Supply Chain Management

17.1 Introduction

One carries inventory for a variety of reasons:

a) protect against uncertainty in demand,
b) avoid high overhead costs associated with ordering or producing small quantities frequently,
c) supply does not occur when demand occurs, even though both are predictable (e.g., seasonal products such as agricultural products, or anti-freeze)
d) protect against uncertainty in supply,
e) unavoidable "pipeline" inventories resulting from long transportation times (e.g., shipment of oil by pipeline, or grain by barge)
f) for speculative reasons because of an expected price rise.

We will illustrate models useful for choosing appropriate inventory levels for situations (a), (b), (c) and (d).

17.2 One Period News Vendor Problem

For highly seasonal products, such as ski parkas, the catalog merchant, L. L. Bean makes an estimate for the upcoming season, of the mean and standard deviation of the demand for each type of parka. Because of the short length of the season, L.L. Bean has to make the decision of how much to produce of each parka type before it sees any of the demand. There are many other products for which essentially the same decision process applies, for example, newspapers, Christmas trees, anti-freeze, and road salt. This kind of problem is sometimes known as the one-period newsvendor problem.

To analyze the problem, we need the following data:

c = purchase cost/unit.
v = revenue per unit sold.
h = holding cost/unit purchased, but not sold. It may be negative if leftovers have a positive salvage value.
p = explicit penalty per unit of unsatisfied demand, beyond the lost revenue.

In addition, we need some information about the demand distribution (e.g., its mean and standard deviation). For the general case, we will presume for any value x:

$F(x)$ = probability demand (D) is less-than-or-equal-to x.

17.2.1 Analysis of the Decision

We want to choose:

S = the stock-up-to level (i.e., the amount to stock in anticipation of demand).

We can determine the best value for S by marginal analysis as follows. Suppose we are about to produce S units, but we ask, "What is the expected marginal value of producing one more unit?" It is:

$$\begin{aligned}
&-c + (v + p) * \text{Prob}\{\text{demand} > S\} - h * \text{Prob}\{\text{demand} \le S\} \\
&= -c + (v + p) * (1 - F(S)) - h * F(S) \\
&= -c + v + p - (v + p + h) * F(S).
\end{aligned}$$

If this expression is positive, then it is worthwhile to increase S by at least one unit. In general, if this expression is zero, then the current value of S is optimal. Thus, we are interested in the value of S for which:

$$-c + v + p - (v + p + h) * F(S) = 0$$

or re-arranging:

$$\begin{aligned}
F(S) &= (v + p - c) / (v + p + h) \\
&= (v + p - c) / [(v + p - c) + (c + h)].
\end{aligned}$$

Rephrasing the last line in words:

Probability of not stocking out should = (opportunity shortage cost)/[(opportunity shortage cost) + (opportunity holding cost)].

This formula is sometimes known as the *news vendor formula*.

Example 1, News vendor with discrete demand distribution:

Suppose L.L. Bean can purchase or produce a parka for \$60, sell it for \$140 during the regular season, and sell any leftovers for \$40. Thus:

$c = 60$,
$v = 140$,
$p = 0$,
$h = -40$.

The opportunity shortage cost is 140 – 60 = 80, and the opportunity holding cost is 60 – 40 = 20. Therefore, the newsvendor ratio is 80/(80 + 20) = 0.8.

To determine S, we must know the demand distribution. First, suppose this is not a big selling parka and we have the distribution in tabular form as follows:

Demand, D:	2	3	4	5	6	7	8	9	11	12	13	14	15
Prob{demand=D}:	.04	.06	.09	.10	.11	.12	.10	.09	.09	.07	.06	.05	.02
Cumulative, $F()$:	.04	.10	.19	.29	.40	.52	.62	.71	.80	.87	.93	.98	1.0

Thus, we should stock S = 11 units.

Example 2, News vendor with Normal distribution:
Suppose we have the same cost structure as before, but this item has a forecasted demand of 1000 and standard deviation of 300. We will make the standard assumption demand is Normal distributed. We must find the number of standard deviations above the mean such that the "left tail" probability is 0.80. From a Normal table, we see this occurs at about $z = .84$. The general form for the stock-up-to point is:

S = mean + (standard deviation) * z,
= 1000 + 300 * .84 = 1252.

It would be nice to know the expected amount of lost sales. The "linear loss function", or *@PSL*() in LINGO gives us this, specifically:

(expected lost sales) = (standard deviation) * *@PSL*(z)
= (standard deviation) * *@PSL*((S – mean)/(standard deviation))
= 300 * *@PSL*(.84)
= 300 * .1120 = 33.6

Alternatively, if we are lazy, we can use LINGO to do all the computations for us with the following model:

```
MODEL:
  ! Newsboy inventory model(NUSBOYGN);
  ! Calculate: optimal order up to stock level, S,
    and re-order point, R, for a
    product with a normally distributed demand;
 DATA:
 MU = 1000; ! Mean demand;
 SD = 300; ! Standard deviation in demand;
  V = 140; ! Revenue/unit sold;
  C = 60; ! Cost/unit purchased;
  P =  0; ! Penalty/unit unsatisfied demand;
  H = -40; !Holding cost/unit left in inventory;
  K = 1000; ! Fixed cost of placing an order;
 ENDDATA
 !----------------------------------------------;
 ! Compute the newsvendor ratio;
 RATIO = ( P + V - C)/( P + V - C + C + H);
 ! Calculate the order up to point, S;
 @PSN( ZS ) = RATIO;
 @FREE( ZS);
 S = MU + SD * ZS;
 ! Compute expected profit of being there, PS;
 ! Note if D = demand, then profit is:
 V * D - V * MAX( 0, D-S) - C * S
 - P * MAX(0,D-S) - H * (S-D) - H*MAX(0,D-S);
 ! Taking expectations and collecting terms...;
 PS = V * MU - C * S - H * ( S - MU)
      - ( V + P + H) * SD* @PSL( ZS) ;
 ! Expected profit at reorder point should differ
   from expected profit at S by fixed order cost, K;
   PR = PS - K;
```

```
   ! Solve for ZR;
   PR = V * MU - C* R - H * ( R - MU)
         - ( V + P + H) * SD* @PSL( ZR) ;
   @FREE( ZR);
   ZR <= ZS; ! Do not want the way over solution;
   ! Finally, compute the re-order point;
   ZR = ( R - MU)/ SD;
 END
```

A solution is:

```
Variable           Value
      MU        1000.000
      SD        300.0000
       V        140.0000
       C        60.00000
       P       0.0000000
       H       -40.00000
       K        1000.000
   RATIO       0.8000000
      ZS       0.8416211
       S        1252.486
      PS        71601.14
      PR        70601.14
       R        1114.215
      ZR       0.3807171
```

The above examples assume the distribution of demand is known. In fact, getting a good estimate of the demand distribution is probably the most challenging aspect of using the news vendor model. The sports clothing retailer, Sport Obermeyer, see Fisher and Raman (1996), derive a demand distribution by soliciting forecasts from six experts. The average of these forecasts is used as the mean of the demand distribution. The standard deviation in the six forecasts is multiplied by an empirically derived adjustment factor (e.g., 1.75) to obtain the standard deviation used in the model. L.L. Bean apparently uses a slightly different approach to estimate the demand distribution for some of its products. A single point estimate forecast for a product is provided by either a single expert, typically a "buyer", or by a consensus forecast from a group. An estimate of the standard deviation is obtained by assuming the coefficient of variation (i.e., standard deviation/mean) remains constant from year to year. The forecast errors from previous years are retained, and thus the coefficient of variation over previous years can be calculated.

17.3 Multi-Stage News Vendor

Advertisements for Lands End Outlet stores typically stress that items being sold in these stores are being sold at a very low price because they are left over from a catalog. Lands End stocked more units than catalog customers were interested in buying. The suggestion is that store customers can benefit from the poor inventory management of the catalog operation.

Similar examples are items carried in a "Christmas" catalog, then offered at a lower price in a "White" sale after the Christmas selling season, and perhaps offered at an even lower price at a third selling opportunity, if there are units still left in stock after the "White" sale. For example, a men's long sleeve plaid shirt that was listed for $36 in a recent L.L. Bean Spring catalog, was listed for $25 in the subsequent Summer Sale catalog. Such multi-level selling situations are here referred to as multi-stage newsvendor problems.

When making the initial stocking decision, one should take into account the selling price and likely demand at each of the downstream levels. It is, in fact, relatively easy to do a fairly accurate analysis of the optimum amount to stock.

For example, suppose a retailer can purchase a particular type of coat from a supplier for \$100. The retailer will offer the garment for sale in the fall selling season for \$225. Any units left over from the fall selling season will be offered in the winter catalog for \$135. Any units still left over at that point will be offered for sale in "outlet" stores for \$95. Demands at the three levels are estimated to have means and standard deviations of:

Label:	**Fall**	**Winter catalog**	**Outlet store**
Mean	1200	300	400
Std.	500	150	190

Intuitively, it seems one should stock about 1200 + 300 = 1500 units because it is profitable to sell the items in the winter catalog at \$135. However, sales in the outlet store are not profitable in retrospect. Can we do a little better than intuition?

Marginal analysis can be used quite nicely in this situation. It goes like this. We are contemplating stocking S units (e.g., 1400 units). Is it, in fact, worthwhile to increase our stocking level to S+1? If yes, we simply repeat until the answer is "no". Let:

D_i = the (not yet seen) demand at stage i, for i = 1, 2, 3;
v_i = the revenue or selling price/unit at level i; and
c = cost/unit.

The expected value of stocking one more unit in the general case is:

$$-c + v_3 * \text{Prob}\{D_1 + D_2 + D_3 > S\} + (v_2 - v_3) * \text{Prob}\{D_1 + D_2 > S\} + (v_1 - v_2) * \text{Prob}\{D_1 > S\}.$$

or in our specific example:

$$-100 + 95 * \text{Prob}\{D_1 + D_2 + D_3 > 1400\} + 40 * \text{Prob}\{D_1 + D_2 > 1400\} + 90 * \text{Prob}\{D_1 > 1400\}.$$

The reasoning behind this expression is as follows:

Stocking the additional item costs \$100.
If the total demand over the three levels is > S, then we clearly can sell the unit for at least \$95.
If the total demand over the first two levels is > S, then we will receive not just \$95, but an additional 135 – 95 = \$40.
If the total demand in the first level is > S, we will receive not just \$135, but an additional 225 – 135 = \$90.

At the optimum, this marginal cost expression should be essentially zero.

If we can assume the demands are Normally distributed at the three levels, then we can compute the expected value of carrying one more unit, and in fact solve for the optimum amount to stock. Note, we do not have to assume the demands are independent at the three levels. The analysis is still correct:

```
MODEL:
  ! Multi-echelon newsboy(MELNUBOY);
  ! Compute stock to carry, S;
  DATA:
  ! The cost/unit of the item;
  C = 100;
  ! Selling price/unit at the first level;
  V1 = 225;
  ! Selling price/unit at the second level;
  V2 = 135;
  ! Selling price/unit at the third level;
  V3 = 95;
  ! Mean demands at the three levels;
  MEAN1 = 1200;
  MEAN2 = 300;
  MEAN3 = 400;
  ! Standard deviations at the three levels;
  SD1 = 500;
  SD2 = 150;
  SD3 = 190;
  ENDDATA
  !----------------------------------------------;
  ! Compute means and s.d. of cumulative demands;
  CUMD3 = MEAN1 + MEAN2 + MEAN3;
  CUMD2 = MEAN1 + MEAN2;
  ! This assumes demands are independent;
  CUMSD3 = (SD1 * SD1 + SD2 * SD2 + SD3 * SD3)^.5;
  CUMSD2 = ( SD1 * SD1 + SD2 * SD2)^.5;
  ! Compute S;
  ! Set to 0 marginal expected value of ordering
  one more unit beyond S, assuming Normal demand.;
  0 = - C
  +          V3 * ( 1 - @PSN(( S - CUMD3)/ CUMSD3))
  + ( V2 - V3) * ( 1 - @PSN(( S - CUMD2)/ CUMSD2))
  + ( V1 - V2) * ( 1 - @PSN(( S - MEAN1)/ SD1));
  ! Compute expected profit;
  !If the demands are D1, D2, and D3, then profit =
  V3* (( D1 + D2 + D3) - MAX( 0, D1+ D2+ D3 - S))
  +( V2 - V3) * (( D1 + D2)- MAX( 0, D1+ D2 - S))
  + ( V1 - V2) * ( D1 - MAX( 0, D1 - S))
  - C * S;
  ! Taking expectations;
  EPROFIT =
  V3 * (CUMD3- CUMSD3* @PSL(( S- CUMD3)/ CUMSD3))
  +(V2- V3)* (CUMD2 -CUMSD2*@PSL((S-CUMD2)/CUMSD2))
  +(V1- V2)* (MEAN1- SD1* @PSL((S- MEAN1)/ SD1))
  - C * S;
END
```

A solution is:

```
   Variable          Value
          C       100.0000
         V1       225.0000
         V2       135.0000
         V3       95.00000
      MEAN1       1200.000
      MEAN2       300.0000
      MEAN3       400.0000
        SD1       500.0000
        SD2       150.0000
        SD3       190.0000
      CUMD3       1900.000
      CUMD2       1500.000
     CUMSD3       555.5178
     CUMSD2       522.0153
          S       1621.628
    EPROFIT       138339.6
```

We see that we should stock substantially more than 1500. Namely, about 1622 units.

17.3.1 Ordering with a Backup Option

One type of "supply chain" agreement used by a number of clothing suppliers (e.g., Liz Claiborne, Ann Klein, and Benetton) is the "backup" supply agreement. A typical agreement is characterized by two numbers, a backup or holdback fraction and a nonuse penalty. Under, say a (.2, .1) backup agreement, a store that orders 100 units of an item from Anne Klein must take delivery of $(1 - .2) \times 100 = 80$ units before the selling season begins. That is, the supplier holds back 20% of the order. During the selling season, the store may additionally request quick delivery on up to $.2 \times 100 = 20$ units at the same price. The store pays a penalty of $.1 \times$ (purchase cost) for each item in the backup for which it does not request delivery. Essentially, the store requests delivery on additional backup items only when it is 100% sure of being able to sell the additional items.

Suppose your store is contemplating a (.2, .1) agreement for a particular item from Anne Klein that has a purchase cost of $50 per unit. You sell it for $160. You were planning to order 100 units of this item. Thus, you will definitely receive 80 and can sell up to 100 if the demand occurs. For any units of the 100 for which you do not take delivery, you must pay $.1 \times \$50 = \$5$. You are now having second thoughts and want to know the marginal value of ordering one more unit of this item.

So, for example, if total demand is greater than the 100, then increasing order size by one is a smart move ($160 – $50). If the demand is less-than-or-equal-to 100, but greater than 80, it is not so smart ($- .1 \times \$50$). If demand is less-than-or-equal-to 80, then it is a dumb move (about – $50, ouch!).

Marginal analysis can be used to determine the best initial order size. We will, in this case, assume any items left over are worthless. Define:

c = cost/unit from the supplier,
v = selling price/unit,
b = holdback fraction,
u = penalty/unit of unused holdback items, stated as a fraction of c,
h = holding cost/unit left over,
D = the (random) demand.

The expected value of ordering one more unit beyond S is:

$$(v - c) * \text{Prob } \{D > S\}$$
$$- u * c * \text{Prob } \{S * (1 - b) < D \leq S\}$$
$$- (c * (1 - b) + u * c * b + h\,(1 - b)) * \text{Prob } \{D \leq S * (1 - b)\}$$

If this expression is positive, S should be increased. At the optimal S, the above expression should be approximately zero. The reasoning behind the three terms is:

If $D > S$, we will take delivery of all units ordered and make a profit of $v - c$ on the extra item ordered.

If $S * (1 - b) < D \leq S$, with or without the extra unit, we take delivery of D units. We have to pay a penalty of $u * c$ on the extra unit ordered, but not delivered.

If $D \leq S * (1 - b)$, we must take delivery of $(1 - b)$ additional units, for which we pay c and incur a holding cost h. We must pay a penalty $u * c$ on the additional units b on which we did not take delivery.

For our example data, suppose D has a Normal distribution with mean 400 and standard deviation 100. The following is a LINGO model for this case:

```
MODEL
  ! Newsboy with a holdback fraction(NUBOYBCK);
  DATA:
  ! Cost/unit;
  C = 50;
  ! Selling price/unit;
  V = 160;
  ! Cost per item left over( <0 for salvage);
  H = - 6;
  ! Holdback fraction;
  B = .2;
  ! Fraction of cost paid on unused units;
  U = .1;
  ! Mean demand;
  MEAN = 400;
  ! Standard deviation in demand;
  SD = 100;
  ENDDATA
  !Set to zero the marginal value of ordering an
  additional unit beyond S;
  ( V - C) * ( 1 - @PSN(( S - MEAN)/ SD))
  - U * C * ( @PSN(( S - MEAN)/ SD)
  - @PSN(( S*( 1 - B) - MEAN)/ SD))
  - (( C + H) * ( 1 - B) + U * C * B )
  * @PSN(( S *( 1 - B) - MEAN)/ SD) = 0;
END
```

A solution is:

```
Variable           Value
       C        50.00000
       V        160.0000
       H       -6.000000
       B       0.2000000
       U       0.1000000
    MEAN        400.0000
      SD        100.0000
       S        493.9043
```

The optimal order quantity is $S = 494$. This means we will take delivery of $0.8 \times 494 = 395$ units, and have the option to receive 99 more if needed.

17.3.2 Safety Lotsize

In the News vendor problem, we have to choose a number (e.g., *S* above) to try to match a random variable (e.g., the demand). A problem that is closely related to the newsvendor problem is the safety lotsize problem. The essential difference is that in the safety lotsize problem, we are given a target number, and we want to choose a distribution, so the associated random variable matches the given target number. The given number is typically a capacity, such as number of seats available on an aircraft, or parking spots in a garage, or the number of units of some product ordered by a customer. In each of these three cases, we may not be able to precisely control how many people show up for a flight, or control how many of the units we put into production turn out to be acceptable. For example, in the manufacture of semiconductor chips, the fraction of acceptable chips in a batch in the early stages of production may be as low as 20%. For airlines, a "no-show" rate of 15% is not unusual. We can, however, affect the number of "good outcomes" by such actions as how many reservations we give out for a flight or a parking lot, or how many chips we start into production. In semi-conductor chip manufacturing, even after considerable production experience is obtained, the yield may still be under 80%.

The following illustrates for the case of the so-called overbooking problem in the airlines. This model does the analysis for three different assumptions about the distribution of the number of customers that do not show up: the Normal distribution, the binomial, and the Poisson.

```
MODEL:
! Safety lot size/ Over booking model(SLOTSIZE);
! Compute S = number reservations to make;
! Keywords: overbooking, safety lotsize, lotsize;
DATA:
! Capacity, e.g., seats available;
   M = 140;
! Prob{ unit is bad or no-show};
   Q = .1;
! Cost per unit put in production;
   C = - 188;
! Penalty per good unit short of target;
   P = 0;
! Holding cost per good unit over target;
   H = 420;
```

```
ENDDATA
!-----------------------------------------------------;
! Model:  Define PROB =;
!  Prob{ Bads <= S - M} = Prob{ Goods >= M};
!  The marginal cost of ordering S+1 rather than S is:
     C - ( 1 - Q) * ( P * ( 1 - PROB) - H * PROB) = 0;
!  Setting to zero, gives;
    PROB = ( P - C/( 1 - Q))/( P + H);
!  Note: can also write as newsboy ratio:
     (P*(1-Q) - C)/(( P*(1-Q) - C) + (C + H*(1-Q)));
! Now determine units to put into production,
  reservations to sell, etc.;
! Binomial(Choose a sample of size SB, where,
  prob{unit is bad} = Q);
    PROB = @PBN( Q, SB, SB - M);
! Poisson approximation;
    PROB = @PPS( Q * SP, SP - M);
! Normal approximation. The .5 improves the
 approximation of the continuous Normal distribution
 to a discrete distribution. The variance of a
 binomial random variable is SN*Q*(1-Q);
    PROB =
       @PSN(( SN - M + .5 - Q * SN)/
            (( SN * Q * ( 1 - Q))^.5));
END
```

The solution is:

```
Variable          Value
       M       140.0000
       Q      0.1000000
       C      -188.0000
       P      0.0000000
       H       420.0000
    PROB      0.4973545
      SB       154.8232
      SP       154.7852
      SN       154.9725
```

Thus, given that 10% of reservation holders do not show up and we have 140 seats to fill, regardless of our distribution assumption, we should sell 155 reservations (and hope exactly 140 customers show up).

17.3.3 Multiproduct Inventories with Substitution

One of the most important issues in inventory management is the consideration of unsatisfied demand, lost sales, or stockouts. When there are multiple related products, unsatisfied demand from one product may be satisfied by some other similar product. General Motors (see for example Eppen, Martin, and Schrage (1989)) has historically used a "diversion matrix" to represent the rate at which unsatisfied demand for one kind of GM car gets satisfied by, or substituted for, some other car. Similar methods have been used in the airlines in choosing capacities for various flights during the day. Here the process may be referred to as "spill" and "recapture". The problem also arises in planning vehicle fleets in the face of uncertain demand for vehicles of various sizes and types. If there is a shortage of small vehicles on a given day, surplus large vehicles may be substituted for the small.

The model below illustrates the essential aspects of the demand diversion inventory model used in the aforementioned GM study. The model is a one-period newsvendor type model, except there are multiple products. Each product has a cost per unit for each unit stocked, a revenue per unit for each unit sold, and a holding cost per unit left over. If there are n products, then shortage costs and the interaction among products is modeled by:

- an n by n diversion matrix that specifies what fraction of the unsatisfied demand of product i may be diverted to and satisfied by product j, and
- an n by n transfer cost matrix that specifies the cost per unit of transferring demand from one product to another.

For example, if a coach class passenger gets upgraded to first class because of lack of space in coach, one can represent this as a sale of a first class seat with the transfer cost being the difference in cost between a first class seat and a coach class seat. This model represents demands by scenarios. Each scenario specifies the demand for all products for that scenario. It is generally convenient to have a n+1st product class that represents the outside world. Demand transferred to it is truly lost.

Example

Multisys, Inc. provides maintenance under contract of desktop computers to industrial firms. Multisys, has just received notice from its disk supplier that it is about to make its last production run for 1 Gig and 2 Gig disk drives. These drives are becoming obsolete as larger capacity drives are becoming available. Multisys still has a large number of computers under maintenance contract that have these 1 and 2 Gig drives. The two drives are plug-compatible physically (i.e., they are the same size and have the same electrical connections). About one third of the computers under contract that have the 1 Gig drive are software incompatible with the 2 Gig drive in that they cannot access or otherwise function with a disk with more than 1 Gig of storage. Otherwise, a 2 Gig drive could be substituted for a 1 Gig drive, and a customer receiving such a substitution would be happy. The 2 Gig drive costs more to Multisys, $200, vs. $140 for the 1 Gig drive. When Multisys replaces a drive, it charges a customer a service charge of either $20 or $30 depending upon whether the original disk is a 1 Gig or a 2 Gig disk. Multisys has enumerated a half dozen scenarios of what its customer requirements might be for replacement disks in the remaining life of their contracts (see the scenarios in the model). If Multisys is short of disks, it will have to buy them on the open retail market, where it expects it would have to pay $190 and $250 respectively for the 1 Gig and 2 Gig drives. Any drive left over after all maintenance contracts have expired is expected to have a salvage value of about $30, regardless of size. How many of each drive should Multisys order from its supplier?

For this problem, the scenario approach introduced in chapter 12 is very convenient. We identify a number of scenarios of what the demands could be. This allows one to have rather arbitrary demand distributions. In particular, demands among the products can be correlated, as is frequently the case in reality. In the example below, we identify a modest six demand scenarios:

```
MODEL:
! Multi-product Newsboy inventory model(NUSBOYML),
  with substitution, diversion, or spill.
 For each product,
  calculate the optimal order up to stock level, S;
SETS:
  PROD/ G1 G2 SPOT/: C, V, H, S;
  PXP( PROD, PROD): FRAC, TC;
  SCEN/1..6/: PROB, PROF;
  SXP( SCEN, PROD): DEM, U, I;
  SXPXP( SCEN, PROD, PROD): T;
ENDSETS
DATA:
! Cost data for 1 Gig and 2 Gig disk drives.
  Third product is outside spot market;
 V =  20   30   0; ! Revenue/unit sold;
 C = 140  200   0; ! Cost/unit stocked;
 H = -30  -30   0; ! Holding cost/unit unused;
!  The diversion matrix. FRAC( PR, PX) = upper limit
    on fraction of product PX unsatisfied demand that
    can be satisfied by product PR;
 FRAC =
       1      0   0  ! Upper limits on;
     .66667   1   0  !  substitution fractions;
       1      1   1; ! Sum over col should be >= 1;
! Transfer costs. TC( PR, PX) = cost per unit of
  satisfying a type PX demand with a type PR product;
 TC =
       0    0    0  ! Cost of transferring;
       0    0    0  ! or substituting one;
     190  250    0; ! product for another;
! The demand scenarios. 3rd product takes care of
  unsatisfied demand;
 DEM = 2100 3300  0
        900 2710  0
       1890 2256  0
       1994 1840  0
       2442 2334  0
       1509 2654  0;
! Prob of each scenario;
! (They are equally likely);
 PROB = .166667  .166667  .166667
        .166667  .166667  .166667;
ENDDATA
```

```
!-----------------------------------------------------;
! Maximize expected profit;
 MAX = @SUM( SCEN( SC): PROB( SC) * PROF( SC));
! For each scenario;
 @FOR( SCEN( SC):
  ! profit =
     revenues - acquisition cost
    - holding cost - transfer costs;
! T( SC, PR, PX) = units of type PX demand satisfied
                   by a type PR product;
   PROF( SC) =
     @SUM( PROD( PR):
           V( PR) * @SUM( PROD( PX): T( SC, PX, PR))
                 - C( PR) * S( PR)
                 - H( PR) * I( SC, PR)
                 - @SUM( PROD( PX):
                     TC( PR, PX) * T( SC, PR, PX)));
   @FREE( PROF( SC));
   @FOR( PROD( PR):
! Stock = inventory left + sent to various products;
   S( PR) = I( SC, PR) + @SUM( PROD( PX):
                                T( SC, PR, PX));
! Directly satisfied + unsatisfied = original demand;
   T( SC, PR, PR) + U( SC, PR) = DEM( SC, PR);
! Unsatisfied demand must be covered from somewhere;
   U( SC, PR) = @SUM( PROD( PX)| PX #NE# PR:
                                 T( SC, PX, PR));
! Cannot send too much to any one place;
    @FOR( PROD( PX)| PX #NE# PR:
       T( SC, PX, PR) <= FRAC( PX, PR) * U( SC, PR);
! In case users find it confusing
   to transfer fractional items;
    @GIN( T( SC, PR, PX));
       );
      );
     );
END
```

When solved, we see the expected net cost is $694,806.4. Hopefully, the maintenance revenues to Multisys are higher than this:

```
Objective value:               -694806.
```

We see Multisys should stock 1508 of the 1 Gig drives and 2334 of the 2 Gig drives. There is at least one scenario in which it must buy 1558 drives on the spot market:

```
Variable           Value
  S( G1)        1508.000
  S( G2)        2334.000
S( SPOT)        1558.000
```

It is interesting to look at the transfers required under each scenario:

```
  T( 1, G1, G1)        1508.000
  T( 1, G2, G2)        2334.000
T( 1, SPOT, G1)         592.000
T( 1, SPOT, G2)         966.000
  T( 2, G1, G1)        900.0000
  T( 2, G2, G2)        2334.000
T( 2, SPOT, G2)         376.000
  T( 3, G1, G1)        1508.000
  T( 3, G2, G1)          78.000
  T( 3, G2, G2)        2256.000
T( 3, SPOT, G1)         304.000
  T( 4, G1, G1)        1508.000
  T( 4, G2, G1)         324.000
  T( 4, G2, G2)        1840.000
T( 4, SPOT, G1)         162.000
  T( 5, G1, G1)        1508.000
  T( 5, G2, G2)        2334.000
T( 5, SPOT, G1)         934.000
  T( 6, G1, G1)        1508.000
  T( 6, G2, G2)        2334.000
T( 6, SPOT, G1)           1.000
T( 6, SPOT, G2)         320.000
```

Notice Multisys plans to go to the spot market under every scenario. In scenarios 3 and 4, surplus 2 Gig drives are substituted for 1 Gig drives.

17.4 Economic Order Quantity

The EOQ model assumes demand is constant over time and any order is satisfied instantly. Define:

D = demand/year,
K = fixed cost of placing an order,
H = holding cost per unit per year.

We want to determine:

Q = quantity to order each time we order.

For any Q chosen, the sum of setup and holding costs is:

$$K * D / Q + h * Q / 2.$$

The minimum of this function occurs when we set:

$$Q = (2 * K * D / h)^{0.5}$$

If we substitute this value for Q back into the cost function, we can find the cost per year if we behave optimally is:

$$(2 * K * D * h)^{0.5}$$

This cost expression illustrates an interesting economy of scale in inventory management with respect to demand volume, D.

Inventory related costs increase with the square root of volume. Thus, if you have two independent facilities, each incurring \$1M per year in inventory related costs, combining them into a single facility will reduce total system costs to \$1.41 M from the original \$2M.

17.5 The Q,r Model

The Q,r model extends the EOQ model with the additional realistic assumptions:

a) there is a positive lead time, and
b) the demand during the lead time is random.

If not for (b), we could trivially extend the EOQ model with the simple observation that we should place our order for the amount Q each time the inventory drops to r = demand during a lead time. Thus, each order will arrive just as inventory hits zero.

If the demand during a lead-time is random, then we will typically wish to increase r slightly to reduce the probability of running out before the order arrives. The Q,r policy is fairly common. For example, Dick Dauch, as Executive Vice President of Worldwide Manufacturing at Chrysler (see Dauch (1993)), used a slight variant of the Q,r model on a wide range of products at Chrysler. Nahmias (1997) gives a thorough introduction to the Q,r model.

17.5.1 Distribution of Lead Time Demand

Define:

L = mean lead time in years,
D = mean demand / year,
sd_L = standard deviation in lead time,
sd_D = standard deviation in demand / year,
MLD = $L * D$ = mean lead time demand.

If demands from one period to the next are independent and identically distributed, then the standard deviation in demand during a lead time, sd_o, is given by:

$$sd_o = (L * sd_D^2 + D * D * sd_L^2)^{0.5}$$

This formula assumes demands, or forecast errors, are independently distributed among periods. In reality, demands (or at least forecast errors) tend to be positively correlated among periods. The result is this formula will typically understate the true standard deviation in lead-time demand or forecast error over the lead-time.

17.5.2 Cost Analysis of Q,r

Define:

$F(r)$ = probability we do not run short in an order cycle if the reorder point is r,
$b(r)$ = expected number of units short in an order cycle if the reorder point is r.

If it is safe to assume lead-time demand has a Normal distribution, then:

$$b(r) = sd_o * @PSL((r - MLD)/ sd_o).$$

For a given Q, and r, the approximate expected cost per year is:

> K * (number of orders per year) + h * (average inventory level) + p * $b(r)$ * (number of orders per year)

The average inventory level can be approximated as follows. On average, the stock level expected at the end of an order cycle (just before an order comes in) is:

$$r - MLD + b(r).$$

The $b(r)$ term is effectively a correction for the fact that $r - MLD$ by itself would be an average over situations, some of which correspond to negative inventory. When inventory is negative, we should not be charging the holding cost h to it (thereby claiming an income rather than a cost). The $b(r)$ term effectively adds back in the negative inventory that would occur when the lead-time demand is greater than r.

When the replenishment order arrives, the stock level is the order quantity Q plus the average quantity in stock at the end of the previous cycle ($r - MLD + b(r)$). The average stock level is the average of these two quantities, $[Q + (r - MLD + b(r)) + (r - MLD + b(r))]/2 = (Q/2 + r - MLD + b(r))$. Note r- $MLD + b(r)$ is the average safety stock in the system.

So, we can write the average cost per year as:

$$= K * D/Q + h * (Q/2 + r - MLD + b(r)) + p * b(r) * D / Q$$

or

$$= (K + p * b(r)) * D / Q + h * (Q/2 + r - MLD + b(r)).$$

This cost expression does not contain a term for inventory in the pipeline (i.e., inventory ordered but not yet on hand). For a given lead time, the average pipeline inventory is a constant equal to $D*L = MLD$. A different holding cost rate may apply to pipeline inventory than to inventory on hand. There may be several reasons why the carrying cost of inventory on order is less than the carrying cost of physical inventory. For example, in the auto industry, a lead time of ten weeks is not unusual for the time from when a dealer places an order with the manufacturer until the order arrives. Of these ten weeks, the first nine weeks might be manufacturing time with only the last week being the time to ship the automobile from the manufacturer to the dealer. The cars are typically shipped FOB (Free On Board/From Our Base) the manufacturer's plant. The dealer thus pays for the car once it ships. So, the dealer incurs inventory carrying costs (e.g., cost of capital, for only one tenth of the lead time).

To minimize the cost, we can either note the similarity of the cost expression to that of the simple EOQ model, or we can differentiate with respect to the parameters and set to zero to get:

$$Q = [2 * D(K + p * b(r))/ h]^{0.5}, \text{ and}$$
$$1 - F(r) = h * Q/(h * Q + p * D), \text{ or}$$
$$F(r) = p * D/ (p * D + h * Q).$$

Note the similarity of the above to the news vendor formula. The intuition is as follows. Suppose we increase the reorder point, r, by one unit. If demand is high during the lead time, then the shortage cost avoided is p. If demand is low, then we simply carried an extra unit in inventory for a cycle, incurring a cost of h * (cycle length) = $h * D/Q$. Using the newsvendor-like arguments, we want to set:

$$F(r) = p/ (p + h*D/Q) = p * D/ (p * D + h * Q).$$

Some textbooks, see Nahmias (1997) for a discussion, using a slightly different approximation to expected inventory level just before an order arrives, get a slightly different expression for $F(r)$, namely:

$$F(r) = (p * D - h * Q)/ (p * D).$$

Both are the result of making approximations to the average inventory level. The latter is intuitively less appealing because, for high values of $h * Q$, it can result in a negative value for $F(r)$. Negative probabilities are hard to comprehend. When $h * Q$ is small relative to $p * D$, then the two expressions result in approximately the same value for $F(r)$. For example, if $p * D = 1.0$ and $h * Q = .05$, then:

$$1/1.05 = 0.952;$$

whereas:

$$(1 - .05)/ 1 = 0.95.$$

Example

When Hewlett-Packard first started supplying printers to Europe, the shipping time from its plant on the west coast of the U.S. to Europe was about five weeks. Suppose the forecasted yearly demand for a certain printer was 270,000 units, with a monthly standard deviation of about 6351. A monthly standard deviation of 6351 implies a monthly variance of 6351 * 6351 = 40333333, a yearly variance (if monthly demands are independent) of 12 * 40333333= 484000000, and a yearly standard deviation of (484000000)^.5 = 22000. The yearly holding cost is $110/printer per year. We allow a separate cost term for pipeline inventory of $5/unit. For example, if we do not have to pay for a product until we receive it, then there would be no charge on pipeline inventory. The penalty for being out of stock when a demand occurs is $200/printer. The fixed cost of placing an order is $300. Suppose the standard deviation in lead-time is two weeks. What should be the re-order point and the re-order quantity? We can have LINGO do all the work for us with the following model:

```
! Q,r inventory model( EOQRMODL);
! Find the order quantity, Q,
   and re-order point, R, for a product with...;
 DATA:
  D = 270000; ! Mean demand / year;
  H = 110; ! Holding cost/unit/year;
  HP=   5; ! Holding cost on pipeline inventory;
  K = 300; ! Fixed order cost;
  P = 200; ! Penalty cost/ unsatisfied demand;
  L = .0962;     ! Lead time in years;
  SDL = .03846; ! S.D. in lead time in years;
  SDD = 22000;  ! S.D. in yearly demand;
 ENDDATA
!-------------------------------------------;
! The Q,R inventory model;
 MLD = L * D;     ! Mean lead time demand;
! s.d. in lead time demand;
SLD=(SDD * SDD * L + D * D * SDL * SDL)^.5;
! Expected cost/ period is ECOST;
 MIN = ECOST;
ECOST = COSTORD + COSTCYC + COSTSFT + COSTPEN + COSTPIPE;
  COSTORD = ( K * D/ Q);
```

```
  COSTCYC = H * Q/2;
  COSTSFT = H*( R - MLD + BR);
  COSTPEN = P * D * BR/ Q;
  COSTPIPE = HP * MLD;
!Expected amount short/cycle.  @PSL() is
  the standard Normal linear loss function;
 BR = SLD * @PSL( Z);
!@PSN()is the standard Normal left tail prob.;
 @PSN( Z) =  P * D /( P * D + H * Q);
 R = MLD + SLD * Z;     ! Reorder point;
! The following are all to help solve it faster;
  Q >= (2*K*D/H)^.5;
  @BND( - 3, Z, 3);
  @FREE( ECOST);   @FREE( R);
  @FREE( COSTORD); @FREE( COSTCYC);
  @FREE( COSTSFT); @FREE( COSTPEN);
  @FREE( Z);       @FREE( BR);
```

Note it breaks the total cost into five components:

1. ordering costs due to the $300 cost of placing an order,
2. cycle inventory due to carrying inventory between order points,
3. holding costs due to carrying safety stock,
4. penalty costs due to being out of stock, and
5. pipeline inventory costs due to product we have paid for, so-called FOB, but not yet received.

It will be interesting to see which of the five is the most significant. A solution is:

```
Variable          Value
       D       270000.0
       H          110.0
      HP            5.0
       K          300.0
       P          200.0
       L         0.0962
     SDL        0.03846
     SDD        22000.0
     MLD        25974.0
     SLD       12425.47
   ECOST      3995220.0
 COSTORD       8991.226
 COSTCYC       495483.0
 COSTSFT      2874377.0
 COSTPEN       486498.0
COSTPIPE       129870.0
       Q       9008.782
       R       52023.54
      BR       81.16215
       Z       2.096463
```

Notice that, of the yearly cost of about $3,995,220, the major component is the safety stock cost of $2,874,377. Comparing the order quantity of 9008 with the yearly demand of 270,000, we can observe this corresponds essentially to ordering every 12 days. The high re-order point, 52,024, relative to the order quantity is because of the long five-week delivery pipeline. Note, five weeks of demand is about 26,000 units.

This model can answer a variety of "what-if" questions regarding how cost is affected by various features of the supply chain. For example, suppose we could switch to a very reliable carrier, so the lead-time is always exactly five weeks. We simply set *SDL* = 0 in the data section as follows:

```
DATA:
  D = 270000; ! Mean demand / year;
  H = 110; ! Holding cost/unit/year;
  HP=   5; ! Holding cost on pipeline inventory;
  K = 300; ! Fixed order cost;
  P = 200; ! Penalty cost/ unsatisfied demand;
  L = .0962;     ! Lead time in years;
  SDL = 0.0;     ! S.D. in lead time in years;
  SDD = 22000;   ! S.D. in yearly demand;
 ENDDATA
```

And get the solution:

```
Variable          Value
       D       270000.0
       H          110.0
      HP            5.0
       K          300.0
       P          200.0
       L         0.0962
     SDL            0.0
     SDD        22000.0
     MLD        25974.0
     SLD       6823.547
   ECOST      2419380.0
 COSTORD       16623.32
 COSTCYC       267997.1
 COSTSFT      1753502.0
 COSTPEN       251387.9
COSTPIPE       129870.0
       Q       4872.674
       R       41892.24
      BR       22.68391
       Z        2.33284
```

So, it looks like the uncertainty in the lead-time is costing us about 3995220 - 2419380 = $1,575,840 a year, most of it in extra safety stock.

We might push the lead time improvement further. Suppose by using airfreight, we could reduce the lead-time from 5 weeks to a reliable 1 week. Our transportation costs will be higher, but how much could we save in inventory related costs? We set $L = 1/52 = .01923$. Thus:

```
DATA:
  D = 270000; ! Mean demand / year;
  H = 110; ! Holding cost/unit/year;
  HP=   5; ! Holding cost on pipeline inventory;
  K = 300; ! Fixed order cost;
  P = 200; ! Penalty cost/ unsatisfied demand;
  L = .01923;     ! Lead time in years;
  SDL = 0.0;     ! S.D. in lead time in years;
  SDD = 22000;   ! S.D. in yearly demand;
ENDDATA
```

Now, the solution is:

```
Variable           Value
       D        270000.0
       H        110.0000
      HP        5.000000
       K        300.0000
       P        200.0000
       L       0.0192300
     SDL       0.0000000
     SDD        22000.00
     MLD        5192.100
     SLD        3050.790
   ECOST        1164946.
 COSTORD        32286.60
 COSTCYC        137982.9
 COSTSFT        863009.1
 COSTPEN        105707.1
COSTPIPE        25960.50
       Q        2508.780
       R        13032.73
      BR        4.911033
       Z        2.570031
```

This looks very promising. Total costs are cut to less than half. Most of the savings, about \$900,000, comes from a reduction in safety stock, about \$400,000 from reduction in pipeline inventory, and about \$100,000 savings each from a reduction in penalty costs and cycle or pipeline stock.

17.6 Base Stock Inventory Policy

If the fixed cost of placing an order is very low relative to the cost of carrying inventory and the cost of being out of stock, then the optimal policy is to reorder one unit whenever a demand occurs. From the Q, r model perspective, the optimal solution has $Q = 1$. Thus, the only decision is R, the reorder point. R is said to be the base stock. An order is placed every time the stock level drops below R. In other words, as soon as demand is observed. Clearly, such a model is interesting only when replenishment lead times are greater than zero. The main tradeoff in the system is between the cost of holding versus the expected cost of backorders or lost sales, just as in the news vendor problem. Base stock policies are very common in aircraft maintenance systems, where spare parts, such as engines, are very valuable relative to the fixed cost of shipping a part to a location where it is needed. Periodic base stock policies are also used for many items in a grocery store. A typical product in a grocery store has a fixed amount of shelf space allocated to it. Early each day, a supplier will stop by the store and fill up the space. The major decision is how much space to allot to each item.

17.6.1 Base Stock — Periodic Review

A slight variation of the basic base stock system is one in which inventory is not checked at every instant, but only periodically. For example, if the product is supplied by ship and the ship arrives only every two weeks, then there is not much benefit in checking inventory constantly. The most typical review period might be weekly (e.g., on Monday mornings after big weekend demand in a retail store). The Newsvendor analysis can then be used to determine the best order-up-to level. Let:

L = lead time in periods,
h = holding cost per unit left in stock at end of period,
p = penalty per unit of demand not satisfied from inventory immediately,
S = pipeline order up to level (also = the reorder point R),
D_t = demand in period t.

We want to determine the best value for S, given known values for L, h, and p, with the D_t's being random variables.

17.6.2 Policy

At the beginning of each period, we observe the pipeline inventory, y, and place an order for $S - y$. Thus, an order placed in period t arrives just before demand occurs in period $t + L$ (but after demand occurs in $t + L - 1$). So, $L = 0$ corresponds to instant delivery. We assume *unsatisfied demand is backlogged*.

17.6.3 Analysis

Just before demand occurs in period $t + L$, the physical inventory available to immediately satisfy demand is:

$$S - \sum_{j=t}^{t+L-1} D_j$$

(e.g., if $L = 0$, the physical inventory is simply S).

If the demands are randomly distributed, let:

$$F(x) = \text{Prob}\ \{ \sum_{j=t}^{t+L} D_j \le x \}$$

Then, by marginal analysis, the expected profit contribution of increasing S by one unit is:

$$p(1 - F(S)) - h\ F(S).$$

Setting this to zero gives:

$$p = (p + h)F(S)$$
or
$$F(S) = p/(p + h)$$

Note, we did not require the assumption that D_t be independently distributed.

The expected holding and shortage cost per period is:

$$E\,[h * \max\,(0, S - \sum_{j=t}^{t+L} D_t) + p * \max\,(0, \sum_{j=t}^{t+L} D_t - S)]$$

$$= E\,[h * (S - \sum_{j=t}^{t+L} D_t) + (p + h) * \max\,(0, \sum_{j=t}^{t+L} D_t - S)]$$

In the case that $\sum_{j=t}^{t+L} D_t$ is Normal with mean μ and s.d. σ, the expected holding and shortage cost can be written as:

$$= h\,(S - \mu) + (p + h)\;\sigma * @PSL\,((S - \mu)\,/\sigma).$$

The lost sales case is very difficult to analyze. The backlogging case as an approximation to the lost sales case will tend to set S too high, understate holding costs, and overstate shortage costs.

Example

An item at a food store is restocked daily. It has a mean demand of 18 units per day with a standard deviation of 4.243. There is a lead-time of two days before an order gets replenished. The holding cost per unit is \$0.005 per day. The shortage penalty per unit is \$0.05 per day.

```
! Base stock policy
  with periodic review and Normal demand(BASESTP)
 DATA:
   H = .005;   ! Holding cost/day;
   P = .05;    ! Shortage penalty/day;
   MEAN = 18;  ! Mean demand/day;
   SD = 4.243;! Std. Dev. in demand/day;
   LEADT = 2; !Lead time in days;
ENDDATA
!--------------------------------------------------;
 MU = LEADT * MEAN;
 SIG = (LEADT * SD * SD)^.5;
 MIN = H * ( S - MU) +
       ( H + P) * SIG * @PSL(( S - MU)/ SIG);
```

The solution is:

```
Optimal solution found at step:        11
Objective value:                0.5399486E-01
Variable           Value        Reduced Cost
       H       0.5000000E-02      0.0000000
       P       0.5000000E-01      0.0000000
    MEAN        18.00000          0.0000000
      SD        4.243000          0.0000000
   LEADT        2.000000          0.0000000
      MU        36.00000          0.0000000
     SIG        6.000508          0.0000000
       S        44.01758          0.8759009E-05
```

So, we should carry a base stock of 44 units and expect holding plus penalty costs to be about \$0.054 per day.

17.6.4 Base Stock — Continuous Review

We say we have continuous review if we review inventory continuously and place an order at any instant that the inventory level drops below the reorder point. Under continuous review, it is convenient to assume demand has a Poisson distribution. In fact, the Poisson distribution is a very appropriate distribution to use for slow moving items. A useful definition of a slow moving item is one for which the mean demand in a period is less than two times its standard deviation. Just as @PSL() is the linear loss function for the Normal distribution, @PPL() is the linear loss function for the Poisson distribution. Arguing much as before, the relevant model for the Poisson distribution is:

```
! Base stock policy
  with continuous review and Poisson demand(BASESTC);
 DATA:
   H = .005;  ! Holding cost/day;
   P = .05;   ! Shortage penalty/day;
   MEAN = 18; ! Mean demand/day;
   LEADT = 2; !Lead time in days;
ENDDATA
!-------------------------------------------------;
   MU = LEADT * MEAN;
   MIN = H * ( S - MU) + ( H + P) * @PPL( MU, S);
```

For this set of data, we get essentially the same result as when the Normal distribution was used:

```
Optimal solution found at step:        66
Objective value:                0.5583237E-01

Variable           Value        Reduced Cost
       H       0.5000000E-02       0.0000000
       P       0.5000000E-01       0.0000000
    MEAN        18.00000           0.0000000
   LEADT        2.000000           0.0000000
      MU        36.00000           0.0000000
       S        43.99994         -0.4514980E-02
```

17.7 Multi-Echelon Base Stock, the METRIC Model

In 1997, the Wall Street Journal reported General Motors (GM) switched to a “distribution center” structure for distributing some of its automobile lines, see Stern and Blumenstein (1996). Previously, all of GM’s finished products were stored at retail car dealers. Under the new system, a significant fraction of cars would be stored at distribution centers (DC) located strategically around the country. Under the old system, if a given dealer did not have the exact style of car desired by a customer, then with high probability that dealer would lose the sale. Even worse for GM, that potential customer might switch to a competing manufacturer’s product.

Under the DC structure, a dealer would typically be able to get, within one day’s time from a nearby DC, the exact car desired by the customer. Under either system, GM must decide:

1) how much inventory to allocate to each dealer.

Under the DC system, GM must also decide:

2) how much inventory to allocate to each DC.

A very similar problem is faced by a large airline. In order to maintain high on-time service, an airline must be able to quickly replace any critical part that fails in an aircraft. For example, the author once had to wait five hours to board a flight because a safety exit chute on the aircraft was accidentally deployed while the aircraft was at the gate. There was a five-hour delay while a replacement chute was flown in from 1500 kilometers away. An airline must decide which parts to stock at which locations around the country. Some high demand parts will be stocked at locations where the demand is likely to occur, and some parts will be stored at centrally located DC's, so they can be quickly flown to low demand cities when demand occurs there.

A key feature of many of these "inventory positioning" problems involving high value items is the appropriate replenishment policy to use as a base stock policy. That is, whenever a demand removes a unit from inventory, an order for a replacement unit is placed immediately. When there are two or more levels in the distribution system (e.g., retail outlets served by one or more DC's), the most widely used model for analyzing this inventory positioning problem is some variation of the METRIC model developed by Sherbrooke (1992) for managing spare parts inventories for the U.S. Air Force. The following model illustrates for the case of five outlets served by a single DC or "depot". In this version, the user specifies, among other parameters, how much stock to carry at the DC and how much stock to allocate over all outlets. The model decides how to best allocate the stock over the outlets and reports the total expected units on backorder.

We look at a situation of how to allocate five units of inventory, say spare engines for an airline, at either a central depot and at each of five demand points:

```
MODEL:
  ! Two level inventory model with possible
  repair at outlet(METRICX);
  ! Compute average units on backorder, TBACK, for
  given limit on depot stock and stock available
  for outlets, using a base stock policy;
  SETS:
  OUTLET/1..5/: ! Each outlet has a...;
  D2OUTL, ! Resupply time from depot to outlet;
  DEM,     ! Demand rate at outlet;
  PREP,  ! Prob item can be repaired at outlet;
  REPT,     ! Repair time at outlet;
  SOUTLET, ! Stock level;
  ERT,     ! Effective resupply time from depot;
  AL;      ! Average level of backlogged demand;
  ENDSETS
  DATA:
  !  Delivery time to outlet from depot(days);
  D2OUTL =  3     7     3     3     9;
  !  Expected demand/day;
  DEM =   .068  .05   .074  .063  .038;
  !  Probability item can be repaired at outlet;
  PREP=   .2    .2    .2    .25   .1;
  !  Repair time at outlet, if repairable;
  REPT=    3     3     3     3     3;
  !  Stock levels to allocate over all outlets;
  SOUTOTL = 5; ! at the depot;
  SDEPOT  = 0; ! Resupply time at depot;
  RDEPOT =  9;
  ENDDATA
```

```
  !-----------------------------------------------;
  ! Compute total demand at depot;
  DEM0 = @SUM( OUTLET: DEM * ( 1 - PREP));
  ! Effective expected wait at depot;
  EWT0 = @PPL( DEM0 * RDEPOT, SDEPOT)/ DEM0;
  @FOR( OUTLET( I):
  ! Estimate resupply time including depot delay;
  ERT( I) = D2OUTL( I) + EWT0;
  ! Expected demand on backorder;
  AL( I) =
  @PPL( DEM( I)* ( 1 - PREP( I)) * ERT( I)
  + DEM( I) * PREP( I) * REPT( I), SOUTLET( I));
  ! Can stock only integer quantities;
  @GIN( SOUTLET( I));
  );
  ! Total expected demand on backorder;
  TBACK = @SUM( OUTLET: AL);
  ! Limit on stock at outlets;
  @SUM( OUTLET( I): SOUTLET( I)) <= SOUTOTL;
  ! Minimize expected backorders;
  MIN = TBACK;
END
```

Case 0: All inventory at outlets:

```
      Variable          Value
        SDEPOT       0.000000
   SOUTLET( 1)       1.000000
   SOUTLET( 2)       1.000000
   SOUTLET( 3)       1.000000
   SOUTLET( 4)       1.000000
   SOUTLET( 5)       1.000000
         TBACK       .9166685
       ERT( 1)       12.00000
       ERT( 2)       16.00000
       ERT( 3)       12.00000
       ERT( 4)       12.00000
       ERT( 5)       18.00000
```

Case 1: One unit at the depot:

```
      Variable          Value
        SDEPOT       1.000000
   SOUTLET( 1)       1.000000
   SOUTLET( 2)       1.000000
   SOUTLET( 3)       1.000000
   SOUTLET( 4)       0.000000
   SOUTLET( 5)       1.000000
         TBACK       .8813626
       ERT( 1)       8.258586
       ERT( 2)       12.25859
       ERT( 3)       8.258586
       ERT( 4)       8.258586
       ERT( 5)       14.25859
```

Case 2: Two units at the depot:

```
         Variable           Value
           SDEPOT        2.000000
      SOUTLET( 1)        0.000000
      SOUTLET( 2)        1.000000
      SOUTLET( 3)        1.000000
      SOUTLET( 4)        0.000000
      SOUTLET( 5)        1.000000
            TBACK        .8683596
          ERT( 1)        5.602399
          ERT( 2)        9.602399
          ERT( 3)        5.602399
          ERT( 4)        5.602399
          ERT( 5)        11.60240
```

Case 3: Three units at the depot:

```
         Variable           Value
           SDEPOT        3.000000
      SOUTLET( 1)        0.000000
      SOUTLET( 2)        1.000000
      SOUTLET( 3)        0.000000
      SOUTLET( 4)        0.000000
      SOUTLET( 5)        1.000000
            TBACK        .9041468
          ERT( 1)        4.094082
          ERT( 2)        8.094082
          ERT( 3)        4.094082
          ERT( 4)        4.094082
          ERT( 5)        10.09408
```

Observe that, from the expected number of units on backorder, the best solution is to put two units at the depot, and one unit at each of locations 2, 3, and 5. This version deals with only a single product and a single DC. See Sherbrooke (1992) for various extensions to this simple version.

17.8 DC With Holdback Inventory/Capacity

Fisher and Raman (1996) describe an approach, called "accurate response" used at the apparel firm, Sport Obermeyer, to help reduce inventories for style goods. The basic setting is two periods with multiple outlets. In the first period, some inventory or production capacity may be held back in order to be allocated in the second period to the outlets that look like they might otherwise run out in the second period. This model has an upper limit, *HBLIM*, on the amount of inventory or capacity that can be held back. In the Sport Obermeyer case, this corresponds to the limited production capacity available at the end of the first period to react to demands observed during the first period. The model allows demands in the second period to be correlated with demands in the first period via the *SHIFT* parameter in the same manner Fisher and Raman (1996) do for Sport Obermeyer. *SHIFT*(*R*, *S*) is the amount by which all demands for retail point (or product) *R*, are shifted up if the demand scenario in the first period was *S*.

```
MODEL:
! Holdback inventory model(HOLDBACK). A central facility
can holdback some inventory or capacity after the first
period to allocate to outlets likely to run out in
the second period;
SETS:
   RETAILP/1..2/: C, V, S1, P1, P2, H1, H2;
   SCENE1/1..4/:;
   SCENE2/1..3/:;
   RXS1( RETAILP, SCENE1): DEM1, SHIFT, Z1, ALLOC;
   RXS2( RETAILP, SCENE2): DEM2;
   RXS1XS2( RETAILP, SCENE1, SCENE2): Z2;
ENDSETS
DATA:
   C = 50 60;  ! Cost/unit for each retail point;
   HBLIM = 80; ! Max available for period 2;
   V = 120 160;! Selling price at each retail point;
   P1=10 11; ! Shortage penalty, lost sales, period 1;
   P2=12 17; ! Shortage penalty, lost sales, period 2;
   H0 = 4;     ! Holding cost per unit in holdback;
   H1 =  5 6;  ! Holding cost at end of period 1;
   H2 = -18 -23; ! At end of period 2;
   DEM1 = 90  60 100 210  ! Demands by scenario;
          50 102  87  45;
   DEM2 = 50  60 100
          70  45  87;
   SHIFT= 12 -10  13  19  ! Shift in period 2 demand;
         -11  14  -8 -15; ! based on period 1 demand;
ENDDATA
!---------------------------------------------------;
! Count number of scenarios;
NS1 = @SIZE( SCENE1);
NS2 = @SIZE( SCENE2);
MAX = REVENUE - PCOST - SHORT1 - SHORT2 - HOLD0 - HOLD1 - HOLD2;
PCOST = @SUM( RXS1( I, K1):
            C( I) * ( S1( I) + ALLOC( I, K1))/NS1;
             );
! Amount ordered = held back + initial allocation;
S = HOLDBK + @SUM( RETAILP( I): S1( I));
```

```
! Limits on amount available for second period;
@BND( 0, HOLDBK, HBLIM);
!  Set Z1 = lost sales in period 1;
@FOR( RXS1( I, K1):
   Z1( I, K1) >= DEM1( I, K1) - S1( I);
);
!  Set Z2 = lost sales in period 2;
@FOR( RXS1XS2( I, K1, K2):
 Z2( I, K1, K2) >= DEM2( I, K2) + SHIFT( I, K1) -
 (S1( I) - DEM1( I, K1) + Z1( I, K1) + ALLOC( I, K1));
);
! Cannot allocate more than was held back;
@FOR( SCENE1( K1):
 @SUM( RETAILP( I): ALLOC( I, K1)) <= HOLDBK;
);
! Compute various average costs;
HOLD0 = H0 * HOLDBK;
HOLD1 = @SUM( RXS1( I, K1):
H1( I)* ( S1( I) - DEM1( I, K1) + Z1( I, K1)))/ NS1;
! If there is a salvage value, HOLD2 could be < 0;
@FREE( HOLD2);
HOLD2 = @SUM( RXS1XS2( I, K1, K2): H2( I) *
 ( S1( I) - DEM1( I, K1) + Z1( I, K1) + ALLOC( I, K1)
 - DEM2( I, K2) - SHIFT( I, K1) + Z2( I, K1, K2)))
 /( NS1 * NS2);
SHORT1 = @SUM( RXS1( I, K1): P1( I) * Z1( I, K1))/NS1;
SHORT2 = @SUM( RXS1XS2( I, K1, K2):
P2( I) * Z2( I, K1, K2))/( NS1 * NS2);
REVENUE = @SUM( RXS1XS2( I, K1, K2): V( I) *
 ( DEM1( I, K1) - Z1( I, K1)
 +  DEM2( I, K2) + SHIFT( I, K1) - Z2( I, K1, K2)))
 /( NS1 * NS2);
END
```

Part of the solution is:

```
Optimal solution found at step:        78
Objective value:                 23496.58
    Variable           Value        Reduced Cost
     REVENUE        44060.00           0.0000000
       PCOST        20600.00           0.0000000
      SHORT1       0.0000000           0.1000000
      SHORT2        49.91667           0.0000000
       HOLD0        320.0000           0.0000000
       HOLD1        745.0000           0.0000000
       HOLD2       -1151.500           0.0000000
           S        406.0000           0.0000000
      HOLDBK        80.00000          -2.000000
      S1( 1)        210.0000           0.0000000
      S1( 2)        116.0000           0.0000000
   Z1( 1, 1)       0.0000000            29.00000
   Z1( 1, 2)       0.0000000            29.00000
   Z1( 1, 3)       0.0000000            21.00000
   Z1( 1, 4)       0.0000000           0.0000000
```

```
     Z1( 2, 1)       0.0000000          28.97500
     Z1( 2, 2)       0.0000000          25.39167
     Z1( 2, 3)       0.0000000          28.97500
     Z1( 2, 4)       0.0000000          26.55833
  ALLOC( 1, 1)       0.0000000          8.000000
  ALLOC( 1, 2)       0.0000000          11.58333
  ALLOC( 1, 3)        3.000000         0.0000000
  ALLOC( 1, 4)        79.00000         0.0000000
  ALLOC( 2, 1)        10.00000         0.0000000
  ALLOC( 2, 2)        80.00000         0.0000000
  ALLOC( 2, 3)        50.00000         0.0000000
  ALLOC( 2, 4)        1.000000         0.0000000
  Z2( 1, 1, 1)       0.0000000          9.500000
  Z2( 1, 1, 2)       0.0000000          9.500000
  Z2( 1, 1, 3)       0.0000000          9.500000
  Z2( 1, 2, 1)       0.0000000          9.500000
  Z2( 1, 2, 2)       0.0000000          9.500000
  Z2( 1, 2, 3)       0.0000000          9.500000
  Z2( 1, 3, 1)       0.0000000          9.500000
  Z2( 1, 3, 2)       0.0000000          9.500000
  Z2( 1, 3, 3)       0.0000000          1.500000
  Z2( 1, 4, 1)       0.0000000          9.500000
  Z2( 1, 4, 2)       0.0000000          8.583333
  Z2( 1, 4, 3)        40.00000         0.0000000
  Z2( 2, 1, 1)       0.0000000          12.83333
  Z2( 2, 1, 2)       0.0000000          12.83333
  Z2( 2, 1, 3)       0.0000000          3.583333
  Z2( 2, 2, 1)       0.0000000          12.83333
  Z2( 2, 2, 2)       0.0000000          12.83333
  Z2( 2, 2, 3)        7.000000         0.0000000
  Z2( 2, 3, 1)       0.0000000          12.83333
  Z2( 2, 3, 2)       0.0000000          12.83333
  Z2( 2, 3, 3)       0.0000000          3.583333
  Z2( 2, 4, 1)       0.0000000          12.83333
  Z2( 2, 4, 2)       0.0000000          12.83333
  Z2( 2, 4, 3)       0.0000000          1.166666
```

The solution recommends ordering 406 units in total and holding back 80 units to allocate out later to the outlets that appear to need it. From the *ALLOC* variables, you can see that if scenario 4 occurs, then retail point 1 gets most of the held back units, otherwise retail point 2 gets most of the held back units.

17.9 Multiproduct, Constrained Dynamic Lot Size Problems

In many production settings, we know demand is not stationary. That is, the demand varies in a predictable way. If we are willing to disregard uncertainty, then efficient methods exist for scheduling production of products over time. One of the earliest occurrences of this problem was the case of a single product with no capacity constraints by Wagner and Whitin (1958). They referred to this problem as the dynamic lot size problem.

We will look at the more general case of multiple products. The most common interaction between products is competition for scarce resources. We first consider the case where each product has essentially the same cost and demand structure as a single product dynamic lot size problem. The products interact by competing for scarce production capacity. This situation can be thought of as a single stage material requirements planning (MRP) problem where production capacities, setup costs, and holding costs are explicitly considered and optimum solutions are sought.

Examples might be the scheduling of production runs of different types of home appliances on an appliance assembly line or the scheduling of different types of automotive tires onto a tire production line. In the applications described by Lasdon and Terjung (1971) and King and Love (1981), several dozen tire types compete for scarce capacity on a few expensive tire molding machines.

The general situation can be described formally by the following example.

17.9.1 Input Data

P = number of products;
T = number of time periods;
d_{it} = demand for product i in period t, for $i = 1, 2, ..., P$; $t = 1, 2, ..., T$;
h_{it} = holding cost charged for each unit of product i in stock at end of period t;
c_{it} = cost per unit of each product i produced in period t;
s_{it} = setup cost charged if there is any production of product i in period t;
a_t = production capacity in period t. We assume the units (e.g., ounces, pounds, grams, etc.) have been chosen for each product, so producing one unit of any product uses one unit of production capacity.

There have been many mathematical programming formulations of this problem. Many of them bad from a computational viewpoint. Lasdon and Terjung (1971) describe a good formulation that has been profitably used for many years at the Kelly-Springfield Tire Company. The following formulation due to Eppen and Martin (1987) appears to be one of the best and enjoys the additional benefit of being moderately easy to describe. The decision variables used in this formulation are:

x_{ist} = fraction of demand in periods s through t of product I, which is produced in period s, where:
$1 \leq s \leq t \leq T;$
= 0 otherwise.
y_{it} = 1 if any product i is produced in period t,
= 0 otherwise.

It is useful to compute the variable cost associated with variable x_{ist}. It is:

$$g_{ist} = c_{is} * (d_{is}\ d_{i,s+1} + ... + d_{it}) + d_{i,s+1} * h_{is} + d_{i,s+2} * (h_{is} + h_{i,s+1}) + ... + d_{it} * (h_{is} + h_{i,s+1} + ... + h_{i,t-1})$$

Similarly, it is useful to compute the amount of production, p_{ist}, in period s associated with using variable x_{ist}:

$$p_{ist} = d_{is} + d_{i,s+1} + ... + d_{it}$$

The objective function can now be written:

$$\text{Min} \sum_{i=1}^{P} \left(\sum_{t=1}^{T} s_{it}\, y_{it} + \sum_{s=1}^{T} \sum_{t=2}^{T} g_{ist} x_{ist} \right)$$

There will be three types of constraints. Specifically:

constraints that cause demand to be met each period for each product,
constraints that, for each product and period, force a setup cost to be incurred if there was any production of that product, and
constraints that force total production to be within capacity each period.

The constraints can be written as:

a) $\sum_{t=1}^{T} x_{i1t} = 1$, for $i = 1, 2, \ldots, P$,

$$\sum_{t=s}^{T} x_{ist} - \sum_{r=1}^{s-1} x_{irs-1} = 0, \qquad \text{for } i = 1, 2, \ldots, P \text{ and } s = 2, 3, \ldots, T$$

b) $y_{is} - x_{iss} - x_{is,s+1} - \ldots - x_{is,T} \geq 0$, for $i = 1, 2, \ldots, P$, and $s = 1, 2, \ldots, T$,

c) $\sum_{i=1}^{P} \sum_{t=s}^{T} p_{ist} + x_{ist}$ for $s = 1, 2, \ldots, T$

All variables are required to be nonnegative. y_{it} is required to be either 0 or 1.

If any of the $d_{it} = 0$, then there must be a slight modification in the formulation. In particular, if $p_{ist} = 0$, then x_{ist} should not appear in constraint set (b). Also, if $p_{ist} = 0$ and $s < t$, then variable x_{ist} may be dropped completely from the formulation.

17.9.2 Example

The parameters of a two-product, constrained, dynamic lotsize problem are as follows:

Demand	**May**	**June**	**July**	**August**	**September**	**October**
Product A:	40	60	100	40	100	200
Product B:	20	30	40	30	25	35
Setup Cost						
Product A:	100	100	150	150	205	200
Product B:	30	40	30	55	45	45
Variable Cost/Unit						
Product A:	5	6	7	8	9	10
Product B:	2	4	4	5	5	5
Unit holding cost/period						
Product A:	1	1	2	2	3	2
Product B:	2	1	1	2	1	2

Production capacity is 200 units per period, regardless of product. Two products can be produced in a period.

An LP/IP formulation for this example appears as follows:

```
MODEL:
!  Two Product Capacitated Lotsizing Problem.
!     Yit = 1 if product i is produced in period t,
!     XAst = 1 if demands in periods s through t are
! satisfied from production in period s, for product
!       A,
!     XBst = 1 etc.  for product B;
MIN  = 100* YA1  +  100* YA2  +  150* YA3
    +  150* YA4  +  205* YA5  +  200* YA6
    +   30* YB1  +   40* YB2  +   30* YB3
    +   55* YB4  +   45* YB5  +   45* YB6
    +  200* XA11 +  560* XA12 + 1260* XA13
    + 1620* XA14 + 2720* XA15 + 5520* XA16
    +  360* XA22 + 1060* XA23 + 1420* XA24
    + 2520* XA25 + 5320* XA26 +  700* XA33
    + 1060* XA34 + 2160* XA35 + 4960* XA36
    +  320* XA44 + 1320* XA45 + 3920* XA46
    +  900* XA55 + 3300* XA56 + 2000* XA66
    +   40* XB11 +  160* XB12 +  360* XB13
    +  540* XB14 +  740* XB15 + 1055* XB16
    +  120* XB22 +  320* XB23 +  500* XB24
    +  700* XB25 + 1015* XB26 +  160* XB33
    +  310* XB34 +  485* XB35 +  765* XB36
    +  150* XB44 +  325* XB45 +  605* XB46
    +  125* XB55 +  335* XB56 +  175* XB66;
!  For product A:
!     If a production lot was depleted in period
! i-1 (the - terms), then a production run of some !sort must be
started in period i (the  + terms);
[A1] + XA11 + XA12 + XA13 + XA14 + XA15 + XA16 = + 1;
[A2] - XA11 + XA22 + XA23 + XA24 + XA25 + XA26 = 0;
[A3] - XA12 - XA22 + XA33 + XA34 + XA35 + XA36 = 0;
[A4] - XA13 - XA23 - XA33 + XA44 + XA45 + XA46 = 0;
[A5] - XA14 - XA24 - XA34 - XA44 + XA55 + XA56 = 0;
[A6] - XA15 - XA25 - XA35 - XA45 - XA55 + XA66 = 0;
!  The setup forcing constraints for A;
[FA1]  YA1 - XA11 - XA12 - XA13 - XA14 - XA15
      - XA16 >=   0;
[FA2]  YA2 - XA22 - XA23 - XA24 - XA25 - XA26 >= 0;
[FA3]  YA3 - XA33 - XA34 - XA35 - XA36 >= 0;
[FA4]  YA4 - XA44 - XA45 - XA46 >= 0;
[FA5]  YA5 - XA55 - XA56 >= 0;
[FA6]  YA6 - XA66 >= 0;
!  Same constraints for product B;
[B1] + XB11 + XB12 + XB13 + XB14 + XB15 + XB16 = + 1;
[B2] - XB11 + XB22 + XB23 + XB24 + XB25 + XB26 = 0;
[B3] - XB12 - XB22 + XB33 + XB34 + XB35 + XB36 = 0;
[B4] - XB13 - XB23 - XB33 + XB44 + XB45 + XB46 = 0;
[B5] - XB14 - XB24 - XB34 - XB44 + XB55 + XB56 = 0;
[B6] - XB15 - XB25 - XB35 - XB45 - XB55 + XB66 = 0;
```

```
!  The setup forcing constraints;
[FB1]  YB1 - XB11 - XB12 - XB13 - XB14 - XB15
     - XB16 >= 0;
[FB2]  YB2 - XB22 - XB23 - XB24 - XB25 - XB26 >= 0;
[FB3]  YB3 - XB33 - XB34 - XB35 - XB36 >= 0;
[FB4]  YB4 - XB44 - XB45 - XB46 >= 0;
[FB5]  YB5 - XB55 - XB56 >= 0;
[FB6]  YB6 - XB66 >= 0;
!  Here are the capacity constraints for each period;
!The coefficent of a variable is the associated lotsize;
[CAP1]  40* XA11 + 100* XA12 + 200* XA13
      + 240* XA14 + 340* XA15  + 540* XA16
      +  20* XB11 +  50* XB12 +  90* XB13 + 120* XB14
      + 145* XB15 + 180* XB16 <=   200;
[CAP2]  60* XA22 + 160* XA23 + 200* XA24
      + 300* XA25 + 500* XA26   + 30* XB22
      +  70* XB23 + 100* XB24 + 125* XB25 + 160* XB26
      <= 200;
[CAP3] 100* XA33 + 140* XA34 + 240* XA35
      + 440* XA36 +  40* XB33   + 70* XB34
      +  95* XB35 + 130* XB36 <=  200;
[CAP4]  40* XA44 + 140* XA45 + 340* XA46
       +  30* XB44 +  55* XB45  + 90* XB46 <= 200;
[CAP5] 100* XA55 + 300* XA56 +  25* XB55
      +  60* XB56 <=   200;
[CAP6] 200* XA66 +  35* XB66 <=   200;
!  Declare the setup variables integer;
@BIN( YA1); @BIN( YA2);
@BIN( YA3); @BIN( YA4);
@BIN( YA5); @BIN( YA6);
@BIN( YB1); @BIN( YB2);
@BIN( YB3); @BIN( YB4);
@BIN( YB5); @BIN( YB6);
END
```

The interpretation of the X_{ijk} variables and the constraint rows 2 through 7 can perhaps be better understood with the picture in the figure below:



The demand constraints, 2 through 7, force us to choose a set of batch sizes to exactly cover the interval from 1 to 6. If an arrow from period 1 terminates at the end of period 3 (production run in period 1 is sufficient for only the first three periods), then another arrow must start at the end of period 3.

If we solve it as an LP (i.e., with the constraints $Y_{it} = 0$ or relaxed to $0 \leq Y_{it} \leq 1$), we get a solution with cost $5,968.125.

When solved as an IP, we get the following solution:

```
Objective Function Value  6030.00000
            Variable          Value
                 YA1           1.000000
                 YA2           1.000000
                 YA6           1.000000
                 YB1           1.000000
                 YB3           1.000000
                 YB5           1.000000
                XA11           0.666667
                XA15           0.333333
                XA25           0.666667
                XA66           1.000000
                XB12           1.000000
                XB34           1.000000
                XB56           1.000000
```

The production amounts can be read off the coefficients of the nonzero X variables in the capacity constraints of the LP. This solution can be summarized as follows:

Product A		Product B	
Period	**Production**	**Period**	**Production**
1	140 $(0.6667 \times 40 + 0.3333 \times 340)$	1	50
2	200 (0.6667×300)	2	0
3	0	3	70
4	0	4	0
5	0	5	60
6	200	6	0

A general, set-based formulation for this example follows:

```
MODEL:
SETS:  ! Multiproduct capacitated lotsizing (CAPLOT);
   TIME ;
   PROD: ST, ! Setup time for product I;
   PT; ! Production time/unit for product I;
   PXT( PROD, TIME):
   D, ! Demand for prod I in period S;
   K, ! Setup cost for prod I in period S;
   C, ! Cost/unit for prod I in period S;
   H, ! Holding cost/unit for prod I, end of period S;
   MAKE, ! Amount to make of I in period S;
   Y; ! = 1 if produce I in period S, else 0;
   PXTXT( PROD, TIME, TIME)| &2 #LE# &3:
   X, ! Fraction of demands in S through T satisfied
        by production in period S;
   VC, ! Variable cost of getting an item from S to T;
   TP; ! Total production in the batch: (I,S,T);
ENDSETS
DATA:
   CAP = 200;   ! Capacity each period;
  PROD= A, B;  ! The products;
   ST = 0  0;  ! Setup time for each product;
   PT = 1  1;  ! Production time/unit for each product;
 TIME= MAY  JUN  JUL  AUG  SEP  OCT;
   D =  40   60  100   40  100  200
        20   30   40   30   25   35;
   K = 100  100  150  150  205  200
        30   40   30   55   45   45;
   H =   1    1    2    2    3    2
         2    1    1    2    1    2;
   C =   5    6    7    8    9   10
         2    4    4    5    5    5;
ENDDATA
!------------------------------------------------------;
@FOR( PXT( I, S):
 VC( I, S, S) = C( I, S);
 TP( I, S, S) = D( I, S);
);
@FOR( PXTXT( I, S, T) | S #LT# T:
! Variable cost of getting product I from S to T;
VC( I, S, T) = VC( I, S, T-1) + H( I, T - 1);
! Total demand for I over S to T;
TP( I, S, T) = TP( I, S, T-1) + D( I, T);
);
MIN = @SUM( PXT( I, T): K( I, T) * Y( I, T))
  + @SUM( PXTXT( I, S, T):
  X( I, S, T) *
     @SUM( PXT( I, J) | S #LE# J #AND# J #LE# T:
                        D( I, J) * VC( I, S, J)));
! Capacity constraints;
@FOR( TIME( S):
 @SUM( PXT( I, S): ST( I) * Y( I, S)) +
```

```
   @SUM( PXTXT( I, S, T):
    TP( I, S, T) * PT( I) * X( I, S, T)) <= CAP;);
 ! Demand constraints;
 @FOR( PROD( I):
 ! First period must be covered;
  @SUM( PXTXT( I, S, T)| S #EQ# 1: X( I, 1, T)) = 1;
 ! For subsequent periods, if a run ended in S-1, then
 we must start a run in S;
  @FOR( TIME( S)| S #GT# 1:
   @SUM( PXT( I, J)| J #LT# S: X( I, J, S - 1)) =
   @SUM( PXTXT( I, S, J): X( I, S, J));
     );   );
 ! Setup forcing constraints;
 @FOR( PXT( I, S):
  @BIN( Y( I, S));
  Y( I, S) >= @SUM( PXTXT( I, S, T):
  @SIGN( TP( I, S, T)) * X( I, S, T));
     );
 ! Compute amount made in each period;
 @FOR( PXT( I, S):
  @FREE( MAKE( I, S));
  MAKE( I, S) =
  @SUM( PXTXT( I, S, T): TP( I, S, T) * X( I, S, T));
     );
 END
```

With comparable solution:

```
Optimal solution found at step:        110
Objective value:                  6030.000
Branch count:                            2

   Variable           Value        Reduced Cost
MAKE( A, 1)        150.0000           0.0000000
MAKE( A, 2)        190.0000           0.0000000
MAKE( A, 6)        200.0000           0.0000000
MAKE( B, 1)        50.00000           0.0000000
MAKE( B, 3)        70.00000           0.0000000
MAKE( B, 5)        60.00000           0.0000000
```

Thus, we make production runs for product A in periods 1, 2, and 6. Production runs for product *B* are made in periods 1, 3, and 5.

17.9.3 Extensions

There are a variety of extensions to this model that may be of practical interest, such as:

Carry-over-setups. It may be a setup cost is incurred in period s only if there was production in period s, but no production in period s - 1. A straightforward, though not necessarily good, way of handling this is by introducing a new variable, z_{it}, related to y_{it} by the relationship: $z_i \geq y_{it} - y_{i,t\text{-}1}$. The setup cost is charged to z_{it} rather than y_{it}.

Multiple machines in parallel. There may be a choice among M machines on which a product can be run. This may be handled by appending an additional subscript m, for m = 1, 2, ..., M, to the x_{ist} and y_{it} variables. The constraints become:

(a′) $\sum_{t=1}^{T} \sum_{m=1}^{M} x_{i1tm} = 1$ for $i = 1, 2, \ldots, P$,

$$\sum_{t=s}^{T} \sum_{m=1}^{m} x_{istm} - \sum_{r=1}^{s-1} \sum_{m=1}^{M} x_{i,r,s\text{-}1,m} = 0 \quad \text{for } i = 1, 2, \ldots, P;$$

$s = 2, \ldots, T.$

(b′) $y_{ism} - x_{issm} - x_{i,s,s+1,m} - \ldots - x_{i,s,T,m} \leq 0$ for $i = 1, 2, \ldots, P$; $s = 1, 2, \ldots, T$; $m = 1, 2, \ldots, m$.

(c′) $\sum_{i=1}^{P} \sum_{t=s}^{T} p_{pstm}\ x_{istm} \leq a_{sm}$ for $s = 1, 2, \ldots, T$, and $m = 1, 2, \ldots, M$.

If the machines are non-identical, then the manner in which p_{istm} is calculated will be machine dependent.

17.10 Problems

1. The Linear Products Company (LPC) of Gutenborg, Iowa, distributes a folding bicycle called the Brompton. Demand for the Brompton over the past year has been at the rate of 5.9 per month, fairly uniformly distributed over the year. The Brompton is imported from a manufacturer in the United Kingdom. For a variety of reasons, including customs processing, small size of the manufacturer, averages of ocean shipping, and getting the shipment from the port of entry to Iowa, the lead time from the manufacturer to LPC is two months. The fixed cost of placing an order, taking into account international phone calls, shipping cost structure, and general order processing is \$200. The cost and selling price per bicycle vary depending upon the features included, but a typical Brompton costs LPC \$500. LPC sells a typical Brompton for \$900. LPC uses a cost of capital of 12% per year.

   a) What order size do you recommend for LPC?
   b) LPC did a statistical analysis of their sales data for the past year and found the standard deviation in monthly demand to be 2.1. LPC estimates a customer who is ready to buy, but finds LPC out of stock, will buy from someone else with probability .8, rather than wait. What reorder point do you recommend for LPC?
   c) LPC did an analysis of their inbound shipments and found that the lead time has a standard deviation of 3 weeks. Extending (b) above, how much is this lead time uncertainty costing LPC?
   d) Suppose LPC could reduce lead time to a reliable one month. Compared to (c) above, how much would this change be worth?

2. A company keeps fleets of vehicles at a number of sites around the country. At each site, the vehicles can be classified into two types, light and heavy. A heavy vehicle costs more per day, but it can do any task that a light vehicle can do. A question of some concern is what mix of vehicles should the company have at each site. If the firm does not have enough vehicles of the appropriate size to meet the demand on a given day, it rents the vehicles. Some cost data were collected on the cost of various vehicle types:

Vehicle type	Daily fixed cost	Daily variable cost(if used)
Owned Light	$32	$40
Owned Heavy	$44	$54
Rented Light	0	$175
Rented Heavy	0	$225

At a particular site, the company collected demand data for the number of vehicles required on each of seven days:

Day	Lights	Heavies
1	6	0
2	3	2
4	8	3
5	2	1
6	4	4
7	1	2

Based on just the above data, what is your recommendation for the number of vehicles to own of each type?

3) A recent option in U.S. tax law is the flexible spending account. If you exploit this option, you are allowed to specify before the year begins, an amount of your salary to be withheld and placed into a "flexible spending" account. During the year, you may withdraw from this account to pay medical expenses that are neither covered by your regular medical insurance, nor deductible on your income tax return as expenses. This account has a "use or lose it" nature in that any money left over in the account at the end of the year is lost to you. You are otherwise not taxed on the amount of money you set aside in this account.

   a) Suppose your tax rate is 35% and you estimate that your uncovered medical expenses during next year have an expected amount of $2400 with a standard deviation of $1100. You are contemplating setting aside S before tax dollars. Write an expression for the expected after tax value of setting aside one more dollar.
   b) How much money should you set aside?
   c) How would you go about estimating the distribution of your medical expenses for next year?